

```

10 REM***MAO, la Memoire Assisitee par Ordinateur***
11 REM***Par Yves SCHLEISS-[09/85]***
20 MODE 2
30 DIM B$(200)
40 WINDOW#1,1,80,1,22
50 WINDOW#2,1,80,24,25
60 WINDOW#3,1,80,23,23
70 CLS
80 FOR h=1 TO 80:PRINT#3,CHR$(154);:NEXT
85 REM***Ecrire/Questionner***
90 CLS#2:PRINT#2,"Terminez vos phrases par:";PRINT#2,"[.] pour Ecrire ou [?] pour Questionner      -      [M] pour Menu"
95 PRINT#1,"          MAO, la Memoire Assisitee par Ordinateur      -[";n;"] lignes":PRINT#1
100 A$="":INPUT#1, "-";A$
110 IF A$="M" OR A$="m" THEN GOSUB 320
120 IF RIGHT$(A$,1)="." THEN GOTO 160
130 IF RIGHT$(A$,1)="?" THEN A$=LEFT$(A$,LEN(A$)-1):GOTO 180
140 PRINT#1, CHR$(7);"Vous avez oublie [.] ou [?]"
150 GOTO 100
160 n=n+1:B$(n)=A$
170 GOTO 100
180 REM***recherche***
190 FOR t=1 TO n

```

```
200 REM***constitution du mot***
210 s=s+1
220 C$=MID$(B$(t),s,1):'prend les lettres dans l'ordre
230 IF C$=CHR$(32) THEN GOTO 270:'teste la fin du mot
240 IF C$="." THEN GOTO 270:'teste la fin de la phrase
250 CT$=CT$+C$:'assemble les lettres pour former le mot
260 GOTO 210
270 REM***comparateur***
280 IF CT$=A$ THEN PRINT#1,B$(t);" #";t:CT$="":GOTO 300
```

```
290 CT$="":IF s=LEN(B$(t)) THEN GOTO 300 ELSE GOTO 210:'initialisation
300 s=0:NEXT t:'initialisation
310 GOTO 100
315 REM***Menu***
320 CLS#1:CLS#2:PRINT#2,"Ecrire/questionner      Charger      Sauver      Lister      Quitter"
330 r$=INKEY$
340 IF r$="E" OR r$="e" THEN GOTO 90
350 IF r$="C" OR r$="c" THEN GOTO 390
360 IF r$="S" OR r$="s" THEN GOTO 480
370 IF r$="Q" OR r$="q" THEN GOTO 570
375 IF r$="L" OR r$="l" THEN GOTO 650
380 GOTO 330
390 REM**Chargement***
400 CLS#2:PRINT#2,"Chargement en cours..."
410 OPENIN "MAOFICHE"
420 INPUT#9,n
430 FOR t=1 TO n
440 INPUT#9,B$(t)
450 NEXT t
460 CLOSEIN
470 GOTO 320
480 REM***Sauvegarde***
490 CLS#2:PRINT#2,"Sauvegarde en cours..."
500 OPENOUT "MAOFICHE"
510 PRINT#9,n
```

```
520 FOR t=1 TO n
530 PRINT#9,B$(t)
540 NEXT t
550 CLOSEOUT
560 GOTO 320
565 REM***Quitter***
570 CLS#2:PRINT#2,"Avez-vous effectue la sauvegarde ? [O/N]"
580 r$=INKEY$
590 IF r$="O" OR r$="o" THEN CLS:END
600 IF r$="N" OR r$="n" THEN GOTO 480
610 GOTO 580
650 REM***liste***
660 CLS#1:CLS#2
670 PRINT#2,"[ESC] 1 fois pour arreter-[ESPACE] pour continuer"
680 FOR t=1 TO 2000:NEXT
690 FOR t=1 TO n
700 PRINT#1,B$(t)
710 NEXT t
720 FOR t=1 TO 2000:NEXT
730 GOTO 320
```